The Next Step in Bump-Mapping

The Scenario:

You want to do bump-mapping. You have a very specific and detailed set of surface normal vectors but don’t have an equation that describes them. Yet you would still like to somehow “wrap” the normal vector field around the object so that you can perform good lighting everywhere.

This is a job for Normal-Maps!
Normal-Mapping is a modeling technique where, in addition to you specifying the color texture, you also create a texture image that contains all of the normal vectors on the object.
How Do You Store a Surface Normal Field in a Texture?

The three components of the normal vector \((nx, ny, nz)\) are mapped into the three color components \((\text{red}, \text{green}, \text{blue})\) of the texture:

\[
\begin{pmatrix} nx \\ ny \\ nz \end{pmatrix} \text{ in the range } -1. \rightarrow 1. \text{ are placed into the texture’s } \begin{pmatrix} \text{red} \\ \text{green} \\ \text{blue} \end{pmatrix} \text{ in the range } 0. \rightarrow 1.
\]

To convert the normal to a color:

\[
\begin{pmatrix} \text{red} \\ \text{green} \\ \text{blue} \end{pmatrix} = \frac{\begin{pmatrix} nx \\ ny \\ nz \end{pmatrix} + 1.}{2.}
\]

To convert the color back to a normal:

\[
\begin{pmatrix} nx \\ ny \\ nz \end{pmatrix} = 2.* \begin{pmatrix} \text{red} \\ \text{green} \\ \text{blue} \end{pmatrix} - 1.
\]
This Gets Us Better Lighting Behavior, While Still Maintaining the Advantages of Bump-Mapping

Ordinary Texture

Normal-Mapping
This Gets Us Better Lighting Behavior, While Still Maintaining the Advantages of Bump-Mapping

Small Specular Shininess

Large Specular Shininess
#version 330 compatibility

out vec3 EC_SurfacePosition;
out vec3 EC_EyePosition;
out vec3 EC_SurfaceNormal;
out vec3 EC_LightPosition;
out vec2 vST;

void main( )
{
    EC_SurfacePosition = (gl_ModelViewMatrix * gl_Vertex).xyz;
    EC_EyePosition = vec3( 0., 0., 0. );
    EC_SurfaceNormal = normalize( gl_NormalMatrix * gl_Normal );
    EC_LightPosition = vec3( 0., 10., 0. );

    vST = gl_MultiTexCoord0.st;

    gl_Position = gl_ModelViewProjectionMatrix * gl_Vertex;
}
#version 330 compatibility

uniform float uKa;
uniform float uKd;
uniform float uKs;
uniform float uShininess;
uniform float uFreq;
uniform sampler2D Color_Map;
uniform sampler2D Normal_Map;

in vec3 EC_SurfacePosition;
in vec3 EC_EyePosition;
in vec3 EC_SurfaceNormal;
in vec3 EC_LightPosition;
in vec2 vST;

void
main()
{
    vec3 P = EC_SurfacePosition;
    vec3 E = normalize( EC_EyePosition - EC_SurfacePosition );
    vec3 N = normalize( gl_NormalMatrix * (2.*texture( Normal_Map, uFreq*vST ).xyz - vec3(1.,1.,1.) ) );
    vec3 L = EC_LightPosition - P;

    vec3 Ambient_Color = uKa * texture( Color_Map, uFreq * vST ).gba;
    float Light_Intensity = 1.;
    L = normalize( EC_LightPosition - P );
    float Diffuse_Intensity = dot( N, L ) * Light_Intensity;
    vec3 Diffuse_Color = uKd * Diffuse_Intensity * texture( Color_Map, uFreq * vST ).rgb;
    float Specular_Intensity = pow( max( dot( reflect( -L, N ), E ), 0. ), uShininess ) * Light_Intensity;
    vec3 Specular_Color = uKs * Specular_Intensity * vec3( 1., 1., 1. );
    gl_FragColor = vec4(Ambient_Color + Diffuse_Color + Specular_Color, 1. );
}